

## Lesson no. 7: Date Function

### Date & Time Functions

To enter a date in Excel, use the "/" or "-" characters. To enter a time, use the ":" (colon). You can also enter a date and a time in one cell.

A1		fx		6/23/2016	
	A	B	C	D	E
1	6/23/2016	6:00	6/23/2016 6:00		
2					

Note: Dates are in US Format. Months first, Days second. This type of format depends on your windows regional settings. Learn more about Date and Time formats.

### Year, Month, Day

To get the year of a date, use the YEAR function.

B1		fx		=YEAR(A1)	
	A	B	C	D	E
1	6/23/2016	2016			
2					

Note: use the MONTH and DAY function to get the month and day of a date.

### Date Function

1. To add a number of days to a date, use the following simple formula.

B1		fx		=A1+5	
	A	B	C	D	E
1	6/23/2016	6/28/2016			
2					

2. To add a number of years, months and/or days, use the DATE function.

B1									
=DATE(YEAR(A1)+4,MONTH(A1)+2,DAY(A1)+9)									
	A	B	C	D	E	F	G	H	I
1	6/23/2016	9/1/2020							
2									

Note: the DATE function accepts three arguments: year, month and day. Excel knows that 6 + 2 = 8 = August has 31 days and rolls over to the next month (23 August + 9 days = 1 September).

## Current Date & Time

To get the current date and time, use the NOW function.

A1					
=NOW()					
	A	B	C	D	E
1	2/23/2017 10:43				
2					

Note: use the TODAY function to enter [today's date](#) in Excel.  
 Hour, Minute, Second

To return the hour, use the HOUR function.

B1					
=HOUR(A1)					
	A	B	C	D	E
1	6:45:17	6			
2					

Note: use the MINUTE and SECOND function to return the minute and second.

## Time Function

To add a number of hours, minutes and/or seconds, use the TIME function.

B1									
=TIME(HOUR(A1)+2,MINUTE(A1)+10,SECOND(A1)+70)									
	A	B	C	D	E	F	G	H	I
1	6:45:17	8:56:27							
2									

Note: Excel adds 2 hours, 10 + 1 = 11 minutes and 70 - 60 = 10 seconds.

## DateDif

To get the number of days, weeks or years between two dates in Excel, use the DATEDIF function. The DATEDIF function has three arguments.

1. Fill in "d" for the third argument to get the number of days between two dates.

A4						
=DATEDIF(A1,A2,"d")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	2623					
5						

Note: =A2-A1 produces the exact same result!

2. Fill in "m" for the third argument to get the number of months between two dates.

A4						
=DATEDIF(A1,A2,"m")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	86					
5						

3. Fill in "y" for the third argument to get the number of years between two dates.

A4						
=DATEDIF(A1,A2,"y")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	7					
5						

4. Fill in "yd" for the third argument to ignore years and get the number of days between two dates.

A4						
=DATEDIF(A1,A2,"yd")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	66					
5						

5. Fill in "md" for the third argument to ignore months and get the number of days between two dates.

A4						
=DATEDIF(A1,A2,"md")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	5					
5						

6. Fill in "ym" for the third argument to ignore years and get the number of months between two dates.

A4						
=DATEDIF(A1,A2,"ym")						
	A	B	C	D	E	F
1	4/18/2009					
2	6/23/2016					
3						
4	2					
5						

Important note: the DATEDIF function returns the number of complete days, months or years. This may give unexpected results when the day/month number of the second date is lower than the day/month number of the first date. See the example below.

A4						
=DATEDIF(A1,A2,"y")						
	A	B	C	D	E	F
1	4/18/2009					
2	4/17/2016					
3						
4	6					
5						

The difference is 6 years. Almost 7 years! Use the following formula to return 7 years.

A4 : X ✓ fx =YEAR(A2)-YEAR(A1)						
	A	B	C	D	E	F
1	4/18/2009					
2	4/17/2016					
3						
4	7					
5						

## Today's Date

To enter today's date in Excel, use the TODAY function. To enter the current date and time, use the NOW function. To enter the current date and time as a static value, use keyboard shortcuts. Today and Now

1. To enter today's date in Excel, use the TODAY function.

A1 : X ✓ fx =TODAY()								
	A	B	C	D	E	F	G	H
1	11/16/2018							
2								

Note: the TODAY function takes no arguments. This date will update automatically when you open the workbook on another date.

2. To enter the current date and time, use the NOW function.

A1 : X ✓ fx =NOW()								
	A	B	C	D	E	F	G	H
1	11/16/2018 14:25							
2								

Note: the NOW function takes no arguments. This time will update automatically whenever the sheet is recalculated. This happens when you make a change to any cell or when you open the workbook. Press F9 to manually recalculate the workbook.

3. To enter the current time only, use NOW()-TODAY() and apply a [time format](#).

	A	B	C	D	E	F	G	H
1	14:26							
2								

Note: dates are stored as numbers in Excel and count the number of days since January 0, 1900. Times are handled internally as numbers between 0 and 1. Visit our page about [date and time formats](#) for more information.

## Static Date and Time

1. To enter the current date as a static value, press CTRL + ; (semicolon).

	A	B	C	D	E	F	G	H
1	11/16/2018							
2								

Note: this date will not change when you open the workbook on another date.

2. To enter the current time as a static value, press CTRL + SHIFT + ; (semicolon).

	A	B	C	D	E	F	G	H
1	2:27 PM							
2								

Note: this time will not change when you make a change to a cell or when you open the workbook.

3. To enter the current date and time as a static value, simply press CTRL + ; (semicolon), enter a space and press CTRL + SHIFT + ; (semicolon).

	A	B	C	D	E	F	G	H
1	11/16/2018 14:28							
2								

## Calculate Age

To calculate the age of a person in Excel, use the [DATEDIF function](#) and the TODAY function. The DATEDIF function has three arguments.

1. Enter the date of birth into cell A2.

A2		fx		4/21/1980					
	A	B	C	D	E	F	G	H	I
1	Date of Birth								
2	4/21/1980								
3									

2. Enter the TODAY function into cell B2 to return today's date.

B2		fx		=TODAY()					
	A	B	C	D	E	F	G	H	I
1	Date of Birth	Today							
2	4/21/1980	10/18/2018							
3									

3. The DATEDIF function below calculates the age of a person.

C2		fx		=DATEDIF(A2,B2,"y")					
	A	B	C	D	E	F	G	H	I
1	Date of Birth	Today	Age						
2	4/21/1980	10/18/2018	38						
3									

Note: fill in "y" for the third argument to get the number of complete years between the date of birth and today's date.

4. Calculate the age of a person without displaying today's date.

B2		fx		=DATEDIF(A2,TODAY(),"y")					
	A	B	C	D	E	F	G	H	I
1	Date of Birth	Age							
2	4/21/1980	38							
3									

5. Calculate the age of a person on a specific date.

=DATEDIF(A2,DATE(2018,1,1),"y")									
	A	B	C	D	E	F	G	H	I
1	Date of Birth	Age							
2	4/21/1980	37							
3									

Note: the DATE function accepts three arguments: year, month and day.

6. Calculate the age of a person in years, months and days.

=DATEDIF(A2,B2,"y") & "y " & DATEDIF(A2,B2,"ym") & "m " & DATEDIF(A2,B2,"md") & "d"									
	A	B	C	D	E	F	G	H	
1	Date of Birth	Today	Age						
2	4/21/1980	10/18/2018	38y 5m 27d						
3									

Note: fill in "ym" for the third argument to ignore years and get the number of months between two dates. Fill in "md" for the third argument to ignore months and get the number of days between two dates. Use the & operator to join strings.



## Time Difference

Calculating the difference between two times in Excel can be tricky. Times are handled internally as numbers between 0 and 1.

Simple Formula

Let's start with a simple formula that calculates the difference between two times in the same day.

1. Simply subtract the start time from the end time.

=B2-A2									
	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	6:00 AM	3:00 PM	9:00						
3									

Note: AM is used for times in the night and morning. PM is used for times in the afternoon and evening. Change the [Time format](#) to display a time with or without AM/PM.

2. Change the number format of the values in cell A2, B2 and C2 to General.



	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	0.25	0.625	0.375						
3									

Note: times are handled internally as numbers between 0 and 1. For example, 6:00 AM is represented as 0.25 (quarter through the day).

### Time Difference Formula

The simple formula shown above doesn't work if the start time is before midnight and the end time is after midnight.

1. Times that are negative show as #####.

	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	10:00 PM	2:30 AM	#####						
3									

2. To clearly see this, change the number format of the values in cell A2, B2 and C2 to General.

	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	0.9166667	0.1041667	-0.8125						
3									

3. The time difference formula below always works.

	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	10:00 PM	2:30 AM	4:30						
3									

Explanation: if the end time is greater than or equal to the start time, simply subtract the start time from the end time. If not, add 1 (full day) to the end time to represent a time on the next day and subtract the start time.

## Time Difference in Hours as Decimal Value

To calculate the difference between two times in hours as a decimal value, multiply the previous formula by 24 and change the number format to General.

1. The formula below does the trick.

C2									
=IF(B2>=A2,B2-A2,B2+1-A2)*24									
	A	B	C	D	E	F	G	H	I
1	Start time	End time	Hours						
2	10:00 PM	2:30 AM	4.5						
3									

## Weekdays

Learn how to get the day of the week of a date in Excel and how to get the number of weekdays or workdays days between two dates.

Weekday function

1. The WEEKDAY function in Excel returns a number from 1 (Sunday) to 7 (Saturday) representing the day of the week of a date. Apparently, 12/18/2017 falls on a Monday.

B1									
=WEEKDAY(A1)									
	A	B	C	D	E	F	G	H	I
1	12/18/2017	2							
2									

2. You can also use the [TEXT function](#) to display the day of the week.

B1									
=TEXT(A1,"dddd")									
	A	B	C	D	E	F	G	H	I
1	12/18/2017	Monday							
2									

3. Or create a [custom date format \(dddd\)](#) to display the day of the week.

A1									
12/18/2017									
	A	B	C	D	E	F	G	H	I
1	Monday								
2									

## Networkdays function

1. The NETWORKDAYS function returns the number of weekdays (weekends excluded) between two dates.

C1		fx		=NETWORKDAYS(A1,B1)					
	A	B	C	D	E	F	G	H	I
1	12/18/2017	12/29/2017	10						
2									

2. If you supply a list of holidays, the NETWORKDAYS function returns the number of workdays (weekends and holidays excluded) between two dates.

C1		fx		=NETWORKDAYS(A1,B1,E1:E2)					
	A	B	C	D	E	F	G	H	I
1	12/18/2017	12/29/2017	8		12/25/2017				
2					12/26/2017				
3									

The calendar below helps you understand the NETWORKDAYS function.

December 2017						
SU	MO	TU	WE	TH	FR	SA
26	27	28	29	30	1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28	29	30
31	1	2	3	4	5	6

3. Dates are stored as numbers in Excel and count the number of days since January 0, 1900. Instead of supplying a list, supply an array constant of the numbers that represent these dates. To achieve this, select E1:E2 in the formula and press F9.

C1		fx		=NETWORKDAYS(A1,B1,{43094;43095})					
	A	B	C	D	E	F	G	H	I
1	12/18/2017	12/29/2017	8						
2									

## Workday function

The WORKDAY function is (almost) the opposite of the NETWORKDAYS function. It returns the date before or after a specified number of weekdays (weekends excluded).

C1									
=WORKDAY(A1,B1)									
	A	B	C	D	E	F	G	H	I
1	12/18/2017	10	1/1/2018						
2									

Note: the WORKDAY function returns the serial number of the date. Apply a [Date format](#) to display the date.

The calendar below helps you understand the WORKDAY function.

December 2017						
SU	MO	TU	WE	TH	FR	SA
26	27	28	29	30	1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28	29	30
31	1	2	3	4	5	6

Again, if you supply a list of holidays, the WORKDAY function returns the date before or after a specified number of workdays (weekends and holidays excluded).